Solution of KKT problem of 27/8/2012

```
> restart:
> with(plots) :
```

Function to minimize

```
> f := x*y ;
```

$$f := x\,y \tag{1}$$

Constraints

```
> g1 := x-y^2+1 ;
  g2 := y^3-x+1 ;
```

$$g1 := x - y^2 + 1$$

$$g2 := y^3 - x + 1 \tag{2}$$

```
> A := contourplot(g1, x=-5..5, y=-5..5,coloring=["white","blue"],
  transparency=0.5,filledregions=true,contours=[0]):
  B := contourplot(g2, x=-5..5, y=-5..5,coloring=["white","red"],
  transparency=0.5,filledregions=true,contours=[0]):
> display(A,B) ;
```



Define the Lagrangian

```
> L := f - mu1 * g1 - mu2 * g2 ;
```

$$L := x\,y - \mu 1\,\left(x - y^2 + 1\right) - \mu 2\,\left(y^3 - x + 1\right) \quad \textbf{(3)}$$

Set the KKT first order nonlinear system

```
> EQ1 := diff(L,x) ;
  EQ2 := diff(L,y) ;
  EQ3 := mu1*g1 ;
  EQ4 := mu2*g2 ;
```

$$EQ1 := y - \mu 1 + \mu 2$$

$$EQ2 := x + 2\,\mu 1\,y - 3\,\mu 2\,y^2$$

$$EQ3 := \mu 1\,\left(x - y^2 + 1\right)$$

$$EQ4 := \mu 2\,\left(y^3 - x + 1\right) \quad \textbf{(4)}$$

Try mu1=mu2=0

```
> subs(mu1=0,mu2=0,[EQ||(1..4)] ) ;
```

$$[y, x, 0, 0] \quad \textbf{(5)}$$

```
> SOL1 := x=0,y=0,mu1=0,mu2=0;
```

$$SOL1 := x = 0, y = 0, \mu 1 = 0, \mu 2 = 0 \quad \textbf{(6)}$$

Try mu1=0, mu2 != 0

```
> EQS := subs(mu1=0,[EQ||(1..4)] ) ;
```

$$EQS := \left[y + \mu 2, x - 3\,\mu 2\,y^2, 0, \mu 2\,\left(y^3 - x + 1\right)\right] \quad \textbf{(7)}$$

```
> EQx := solve( EQS[4], {x}) ; EQmu2 := solve( EQS[1], {mu2} ) ;
  subs( EQmu2, subs( EQx, EQS[2] )); EQy := solve( %, {y} ) ;
```

$$EQx := \left\{x = y^3 + 1\right\}$$

$$EQmu2 := \left\{\mu 2 = -y\right\}$$

$$4\,y^3 + 1$$

$$EQy := \left\{y = -\frac{1}{2}\,2^{1/3}\right\}, \left\{y = \frac{1}{4}\,2^{1/3} - \frac{1}{4}\,\mathrm{I}\,\sqrt{3}\,2^{1/3}\right\}, \left\{y = \frac{1}{4}\,2^{1/3} + \frac{1}{4}\,\mathrm{I}\,\sqrt{3}\,2^{1/3}\right\} \quad \textbf{(8)}$$

```
> SOL2 := op(subs(EQy[1],EQx)),op(EQy[1]),mu1=0,op(subs(EQy[1],EQmu2)
  );
```

$$SOL2 := x = \frac{3}{4}, y = -\frac{1}{2}\,2^{1/3}, \mu 1 = 0, \mu 2 = \frac{1}{2}\,2^{1/3} \quad \textbf{(9)}$$

Try mu2=0, mu1 != 0

```
> EQS := subs(mu2=0,[EQ||(1..4)] ) ;
```

$$EQS := \left[y - \mu 1, x + 2\,\mu 1\,y, \mu 1\,\left(x - y^2 + 1\right), 0\right] \quad \textbf{(10)}$$

```
> EQx := solve( EQS[3], {x}) ; EQmu1 := solve( EQS[1], {mu1} ) ;
  subs( EQmu1, subs( EQx, EQS[2] )); EQy := solve( %, {y} ) ;
```

$$EQx := \left\{x = y^2 - 1\right\}$$

$$EQmu1 := \left\{\mu 1 = y\right\}$$

$$3\,y^2 - 1$$

$$EQy := \left\{y = \frac{1}{3}\,\sqrt{3}\right\}, \left\{y = -\frac{1}{3}\,\sqrt{3}\right\} \quad \textbf{(11)}$$

```
> SOL3 := op(subs(EQy[1],EQx)),op(EQy[1]),mu2=0,op(subs(EQy[1],EQmu1));
```

$$SOL3 := x = -\frac{2}{3}, y = \frac{1}{3}\sqrt{3}, \mu 2 = 0, \mu 1 = \frac{1}{3}\sqrt{3} \quad \textbf{(12)}$$

```
> SOL4 := op(subs(EQy[2],EQx)),op(EQy[2]),mu2=0,op(subs(EQy[2],EQmu1));
```

$$SOL4 := x = -\frac{2}{3}, y = -\frac{1}{3}\sqrt{3}, \mu 2 = 0, \mu 1 = -\frac{1}{3}\sqrt{3} \quad \textbf{(13)}$$

SOL4 must be discarded because m1 < 0

Try mu2!=0, mu1 != 0

```
> EQS := [EQ||(1..4)] ;
```

$$EQS := \left[y - \mu 1 + \mu 2, x + 2\,\mu 1\, y - 3\,\mu 2\, y^2, \mu 1\,\left(x - y^2 + 1\right), \mu 2\,\left(y^3 - x + 1\right)\right] \quad \textbf{(14)}$$

```
> subs( solve( EQS[3],{x} ),  EQS[4]) ; SOLy := solve( %, {y} ) ;
```

$$\mu 2\,\left(y^3 - y^2 + 2\right)$$

$$SOLy := \{y = -1\}, \{y = 1 - I\}, \{y = 1 + I\} \quad \textbf{(15)}$$

```
> subs(SOLy[1],EQS[3] ) ; SOLx := x = 0 ;
```

$$\mu 1\, x$$

$$SOLx := x = 0 \quad \textbf{(16)}$$

```
> SYSMU := subs( SOLy[1], subs( SOLx, EQS[1..2] )) ;
```

$$SYSMU := \left[-1 - \mu 1 + \mu 2, -2\,\mu 1 - 3\,\mu 2\right] \quad \textbf{(17)}$$

```
> solve( SYSMU, [mu1,mu2] ) ;
```

$$\left[\left[\mu 1 = -\frac{3}{5}, \mu 2 = \frac{2}{5}\right]\right] \quad \textbf{(18)}$$

Discard the solution because mu1 < 0

Discussion of the first point

```
> Hess := <<diff(L,x,x),diff(L,x,y)>|<diff(L,x,y),diff(L,y,y)>> ;
```

$$Hess := \begin{bmatrix} 0 & 1 \\ 1 & 2\,\mu 1 - 6\,\mu 2\, y \end{bmatrix} \quad \textbf{(19)}$$

```
> subs( SOL1, [g1,g2] ) ;
```

$$[1, 1] \quad \textbf{(20)}$$

No constraint is active, the kernel is the whole R^2

```
> subs(SOL1,Hess) ; LinearAlgebra[Eigenvalues](%) ;
```

$$\begin{bmatrix} 0 & 1 \\ 1 & 0 \end{bmatrix}$$

$$\begin{bmatrix} 1 \\ -1 \end{bmatrix} \quad \textbf{(21)}$$

Discussion of the second point

```
> subs( SOL2, [g1,g2] ) ;
```

(22)

$$\left[\frac{7}{4}-\frac{1}{4}\,2^{2/3},0\right] \qquad (22)$$

The second constraint is active

```
> gradg2 := subs(SOL2,<diff(g2,x)|diff(g2,y)>) ;
```

$$gradg2 := \begin{bmatrix} -1 & \frac{3}{4}\,2^{2/3} \end{bmatrix} \qquad (23)$$

```
> Z := op(LinearAlgebra[NullSpace](gradg2)) ;
```

$$Z := \begin{bmatrix} \frac{3}{4}\,2^{2/3} \\ 1 \end{bmatrix} \qquad (24)$$

```
> Hess2 := subs(SOL2,Hess) ;
```

$$Hess2 := \begin{bmatrix} 0 & 1 \\ 1 & \frac{3}{2}\,2^{2/3} \end{bmatrix} \qquad (25)$$

```
> LinearAlgebra[Transpose](Z).Hess2.Z ;
```

$$3\,2^{2/3} \qquad (26)$$

The point is a minimum, satisfy sufficient condition.

Discussion of the last point

```
> subs( SOL3, [g1,g2] ) ;
```

$$\left[0, \frac{1}{9}\sqrt{3}+\frac{5}{3}\right] \qquad (27)$$

The first constraint is active

```
> gradg1 := subs(SOL3,<diff(g1,x)|diff(g1,y)>) ;
```

$$gradg1 := \begin{bmatrix} 1 & -\frac{2}{3}\sqrt{3} \end{bmatrix} \qquad (28)$$

```
> Z := op(LinearAlgebra[NullSpace](gradg1)) ;
```

$$Z := \begin{bmatrix} \frac{2}{3}\sqrt{3} \\ 1 \end{bmatrix} \qquad (29)$$

```
> Hess3 := subs(SOL3,Hess) ;
```

$$Hess3 := \begin{bmatrix} 0 & 1 \\ 1 & \frac{2}{3}\sqrt{3} \end{bmatrix} \qquad (30)$$

```
> LinearAlgebra[Transpose](Z).Hess3.Z ;
```

$$2\sqrt{3} \qquad (31)$$

The point is a minimum, satisfy sufficient condition.